HRP 259 SAS LAB TWO


Lab Two: Getting data into SAS, looking at data, analyzing 2x2 Tables in SAS
Lab Objectives

After today’s lab you should be able to:

1. Directly enter a new dataset into SAS EG. 

2. Input grouped data into SAS EG.
3. Analyze a simple 2x2 table in SAS or SAS EG using PROC FREQ/Table Analysis.

4. Import data into SAS EG from an excel spreadsheet.

5. Understand the different variable types in SAS (character, numeric, datetime, currency).

6. Generate numerical and pictorial summaries of data using PROC UNIVARIATE/Distribution Analysis.

7. Calculate Odds Ratios and 95% confidence intervals in SAS.

8. Understand the concept of a SAS MACRO.

SAS PROCs                                     

SAS EG equivalent
PROC FREQ                             

Describe(Table Analysis
PROC UNIVARIATE


Describe(Distribution Analysis…
 LAB EXERCISE STEPS:

Follow along with the computer in front…

1.  Open SAS EG: From the desktop( double-click “Applications”( double-click SAS EG icon. 

2. Click on “New Project”

3. Click on File(New(Data to create a new dataset. This dataset will contain 4 variables: cancer (character variable), IsCase (numeric variable), IsSmoker (numeric variable), and counts (numeric variable). 

Name the dataset “lab2” and store the dataset in the work library (this library temporarily stores datasets). Then click Next.

Name a variable “cancer” that is a character variable:

Name a variable “IsCase” that is a numeric variable.

Name a variable IsSmoker that is a numeric variable. Name a variable counts that is a numeric variable. Delete variables E and F, which we will not use. Then click Finish.

4. Directly type the following data values into the dataset. Right click on empty rows to delete them.

5. FYI, the code to enter the same data is the following:
data lab2;

input iscase issmoker counts;

datalines;

1 1 100

1 0 50

0 1 50

0 0 100

;

run;
6. To put these data into a two-by-two table and calculate odds ratios and risk ratios, start a new program in SAS. Program(New Program and then type the following code:



proc freq data=work.lab2 order=data;

tables IsSmoker*IsCase/ measures norow nopercent;


weight counts;
run;


The results should show the 2x2 table as well as odds ratios, risk ratios, and 95% confidence intervals:

	Estimates of the Relative Risk (Row1/Row2)

	Type of Study
	Value
	95% Confidence Limits

	Case-Control (Odds Ratio)
	4.0000
	2.4749
	6.4649

	Cohort (Col1 Risk)
	2.0000
	1.5529
	2.5758

	Cohort (Col2 Risk)
	0.5000
	0.3882
	0.6440


7. To get the same output using point and click, use Table Analysis. Select Describe(Table Analysis:

In the Data screen, drag “IsSmoker” and “IsCase” to make them the Table variables and “counts” to make it the frequency (weighting) variable.

Click on “Tables” in the left-hand menu. In the Tables screen, drag and drop IsCase to make it the row variable, and drag IsSmoker to make it the column variable.

Click on Table Statistics in the left-hand menu. In this screen, check the box labeled “measures” to ask for relative risk measures. Then click Run.

8. Examine the output. Some nice graphics are created in addition to what we generated with our custom code. 

9. Note that the 2x2 Table is “backwards”—that is 0’s precede 1’s. This is the default in SAS. To fix this we can alter the automatically generated code. Click on the Code tab and make the following change: (Click YES when it asks if you want to make a modifiable copy of the code).
PROC FREQ DATA = WORK.SORTTempTableSorted 

ORDER=INTERNAL DATA;


WEIGHT counts;


TABLES IsSmoker * IsCase /



NOROW


NOPERCENT


NOCUM


MEASURES


SCORES=TABLE



ALPHA=0.05;
10. If all you want to do is get ORs and confidence limits from a 2x2 Table, it’s faster to just use SAS as a calculator to do this (since we know the formulas). We’ll do this at the end of class.

11. Download the class dataset from the class website.  Goto: www.stanford.edu/~kcobb/courses/hrp259
Scroll down to “Extras” and right-click on “Class Data for Lab 2.”  Save data as “classdata2011.xls” to your desktop

12. Import the data into SAS: File(Import Data(browse to find the classdata2011.xlsx dataset on your desktop.

13. In the fist screen, Browse to find the work library. We will store the new SAS dataset classdata in the Work library for now. A library is a reference to a physical folder where datasets are stored. The work library is a folder within SAS. After putting the dataset in the work library, you can press “Finish” immediately. 

14. Use point-and-click to examine the distributions of several continuous variables.
Describe(Distribution Analysis

In the Data screen, drag several variables to “Analysis variables” (

In the left-hand menu, click on Plots (or Appearance). Check the Histogram Plot box. Then click Run.

15. Try also writing a program to produce these same statistics and histograms. Program(New Program
proc univariate data= classdata2011;



pattern1 color = blue;



var coffee milk alcohol;



histogram / endpoints = 0 to 30 by 2;

run;

Compare with the code automatically generated by SAS from Distribution Analysis. 

16. For categorical and binary variables, ask for one-way frequencies:

Drag and drop relevant variables to Analysis variables. Then click Run. 

17. FYI, the equivalent custom code for generating one-way frequencies in SAS is:

   
proc freq data= classdata2011 ;                                                                                                       
     

tables IsCommuter Varsity BookSmart;

   
run;
18. Create a new program (Program(New Program) and enter the following code:


proc format;

value YesNo

0="no"
1=" yes";

run;
data classdata2011;

set classdata2011;

if engSAT>720 then Higheng=1; *above median value=high;
else if engSAT>=0 then Higheng=0;

run;

data classdata2011;

set classdata2011;

if WritingLove>5 then ProWrite=1; *above 5 = positive about math;
else if Writinglove>0 then ProWrite=0;run;


proc freq data= classdata2011  order=formatted ;                                                                                                       

tables HighEng*ProWrite /measures norow nopercent;

format ProWrite yesno.;

format HighEng yesno.;

run;
Results:
	Table of HighEng by ProWrite

	 
	ProWrite
	Total

	
	yes
	no
	

	HighEng
	 
	5
	4
	9

	yes
	Frequency
	
	
	

	
	Col Pct
	62.50
	33.33
	 

	no
	Frequency
	3
	8
	11

	
	Col Pct
	37.50
	66.67
	 

	Total
	Frequency
	8
	12
	20

	Frequency Missing = 4


	Estimates of the Relative Risk (Row1/Row2)

	Type of Study
	Value
	95% Confidence Limits

	Case-Control (Odds Ratio)
	3.3333
	0.5148
	21.5839

	Cohort (Col1 Risk)
	2.0370
	0.6592
	6.2943

	Cohort (Col2 Risk)
	0.6111
	0.2705
	1.3808

19. Generalize the above code to a MACRO! A SAS macro is just a function. You can save it for future use, to avoid repetitive coding. Program(New Program
Enter the following macro to calculate upper and lower confidence limits for any 2x2 table.  The user enters the desired level of confidence (e.g., 95%, 99%, etc.) and the cell sizes from the 2x2 table (cells a-d). The macro calculates the point estimate and confidence limits for the given 2x2 table and enters the results into the SAS LOG. 


· A % sign in SAS denotes a macro name.

· In SAS, a variable bracketed by & and . (e.g., &a.) denotes a macro variable (entered into the macro by the user).

 /**MACRO to calculate XX% confidence limits for an odds ratio 

for a given confidence level (entered as a whole number, eg “95”) 

and the 2x2 cell sizes: a,b,c,d, where a is the diseased, exposed 

cell**/

%macro oddsratio (confidence,a,b,c,d); *enter confidence 

percent as a whole number, e.g. "95";
data _null_;

OR=&a.*&d./(&b.*&c.);

lnOR=log(OR);

error=sqrt(1/&a.+1/&b.+1/&c.+1/&d.);

Z=-probit((1-&confidence./100)/2); *gives left hand 

Z score, multiply by negative;
lower=exp(lnOR-Z*error);

upper=exp(lnOR+Z*error);

put OR;

put lower;

put upper;

run;

%mend oddsratio;

/**Invoke MACRO using data from math example and ask for 95% confidence limit**/

%oddsratio(95, 5, 4, 3, 8);


SAS LOG should contain:

3.3333333333

0.5147700169

21.584612053

/**Invoke MACRO using data for a different 2x2 Table, such as the one we started the lab with! **/

%oddsratio(95, 60, 20, 40, 50);

Notes: For today, we have not bothered to save either dataset that we have created (lab2 or classdata2011). When we close SAS, these datasets will be lost (though they can easily be regenerated by re-running the saved project). Next week we will learn about libraries, places where datasets can be stored permanently…

PROC FREQ gives frequency tables and related statistics. 





The probit function returns the Z score associated with a given area under a normal curve.





When creating a macro, it’s important to include detailed comments that instruct a new user on how to use your macro.

PROC FORMAT allows the user to define a format for any variable. Instead of displaying the values “0” and “1”, SAS will display the values “no” and “ yes” wherever a “binary” format is called.





When calling a format, always add a period to the format name. 








FORMAT only affects the display of variables. 








Create binary variables from continuous variables using if-then statements.








Asks for  measures of association, such as odds ratios. 





Orders the values alphabetically by their formatted values (_Yes comes first because of the space)








Tells SAS to order the values in the 2x2 table according to how they appear in the dataset. So that 1’s precede 0’s. 





Asks for measures of relative risk, e.g., risk ratios and odds ratios. 





Use SAS as a calculator to get these numbers:


data _null_;


OR=5*8/(4*3);


lnOR=log(OR);


error=sqrt(1/5+1/4+1/3+1/8);


Z=1.96; 


lower=exp(lnOR-Z*error);


upper=exp(lnOR+Z*error);


put OR;


put lower;


put upper;


run;


In the Log:


3.3333333333


0.5147700169


21.584612053








Norow suppresses row percents in the Table, and nopercent suppresses cell percents. 





Weights the observations by the count variable. Otherwise, n=4.





Options appear after the forward slash in SAS. 











Formats can also be created using poin-and-click: Data(Create Format 

